

**Exercice 1-1 : Tracé d'une spirale**

Le programme listé ci-dessous permet le tracé d'une figure qui est une version simplifiée de la figure 1.3 du livre. Le dialogue initial permet de saisir les valeurs des paramètres  $\alpha$  et  $\beta$ , tout en vous permettant de conserver les « valeurs par défaut » 0.6 et 7. Parmi les instructions de tracé en deux dimensions disponibles en Scilab, `plot2d` et `plot` sont les plus courantes; la seconde, imitée de Matlab, est un peu plus commode. Elles produisent des résultats identiques, pour des jeux de paramètres équivalents. D'après l'aide de Scilab : « `subplot(m,n,p)` divise la fenêtre graphique courante en une matrice  $m \times n$  de sous-fenêtres et sélectionne la  $p$ -ième sous-fenêtre comme emplacement de dessin par défaut. Le numéro de la sous-fenêtre est compté ligne par ligne, c'est-à-dire que l'emplacement  $(i,j)$  de la matrice porte le numéro  $(i-1)*n + j$ . »

```
//Exercice 1-1: tracé d'une spirale , version simple
clear; //mise à zéro de toutes les variables du programme
clf(); //vide les fenêtres graphiques
t = linspace(0,5,200)'; //vecteur colonne des valeurs de t

//dialogue pour définir les paramètres
txt = ["amortissement","pulsation"];
rps = x_mdilog("choisissez vos paramètres",txt,["0.6","7"]);
alf = evstr(rps(1)); bet = evstr(rps(2));

x = exp(-alf*t).*cos(bet*t); //.* est l'opérateur de
//multiplication élément par élément de deux vecteurs
y = exp(-alf*t).*sin(bet*t);

subplot(2,2,1);
//y(t)
plot(t,y,"r");

subplot(222);
//tracé de la spirale
plot2d(x,y,style=2);

subplot(2,2,4);
//t(x)
plot(x,t,"g")
```

Le listing suivant est la version complète, qui affiche les axes, les légendes et les graduations comme dans le texte. Il est assez compliqué du fait du recours à de nombreuses instructions graphiques. Une bonne façon de s'initier à ce jargon consiste à introduire progressivement, dans le programme de base, des instructions du programme complet et d'observer leur effet.

```
//Exercice 1-1: tracé d'une spirale
clear; //mise à zero de toutes les variables du programme
clf(); //vide les fenetres graphiques
t = linspace(0,5,200)'; //vecteur colonne des valeurs de t

txt = ["amortissement","pulsation"];
rps = x_mdilog("choisissez vos paramètres",txt,["0.6","7"]);
alf = evstr(rps(1)); bet = evstr(rps(2));
```

```

x = exp(-alf*t).*cos(bet*t);
y = exp(-alf*t).*sin(bet*t);

subplot(2,2,1);
//y(t)

plot(t,y,"r");
plot(t,0,"k"); //axe horizontal
a=gca();
a.axes_visible="on";
a.box="on";
a.data_bounds=[0,-1;5,1];
//les positions et les intitulés des graduations
//doivent être initialisés ensemble
//à l'intérieur d'une structure comme temp ou a.x_ticks
ticy = -1:1:1;
ticyl = string(ticy);
ticx = 0:1:5;
ticxl = string(ticx);
temp = a.x_ticks;
temp.locations=ticx;
temp.labels=ticxl;
a.x_ticks = temp;
temp = a.y_ticks;
temp.locations=ticy;
temp.labels=ticyl;
a.y_ticks = temp;
xlabel("t"); ylabel("y");

subplot(222);
//tracé de la spirale

plot2d(x,y,style=2);
a=gca();
a.axes_visible="on";
a.data_bounds=[-1,-1;1,1];
a.isoview="on";
a.sub_ticks=[0,0];
a.box="on";
xlabel("x"); ylabel("y");
horiz = -1:0.1:1;
plot(horiz,0,"k"); //axe horizontal;
plot(zeros(horiz),horiz); //axe vertical

subplot(2,2,4);
//t(x)

plot(x,t,"g")
a=gca();
a.axes_visible="on";
a.box="on";
a.data_bounds=[-1,0;1,5];
xlabel("x"); ylabel("t");
temp = zeros(t);
plot(temp,t,"k"); //axe vertical

```